

# AIDA - an Air Infiltration Development Algorithm

Martin Liddament

Head of Air Infiltration and Ventilation Centre



## Introduction

AIDA is a basic infiltration calculation procedure intended for the calculation of air change rates in single zone enclosures. It also determines flow rates for up to 10 user defined openings and calculates wind and stack pressures. This program is very easy to use but, nevertheless, provides an accurate solution to the air infiltration flow balance equation. As its name suggests this is a development algorithm which may be readily adapted to include additional flow paths, mechanical ventilation and many other features. In less than 60 lines of computer code, AIDA is capable of solving almost any single zone problem on almost any level of computer. This algorithm uses the concepts outlined in Chapter 3 of the AIVC's Calculation Techniques Guide (Liddament 1986).

## 1 Theoretical Outline

Solution is based on the iterative balancing of the flow equation given by:

$$Q = \sum_{j=1}^I C_j (\rho_j)^{n_j} = 0$$

where  $j$  = flow path number  
 $I$  = total number of flow paths  
 $C_j$  = flow coefficient of flow path  $j$   
 $n_j$  = flow exponent of flow path  $j$   
 $\Delta p_j$  = pressure difference across flow path  $j$

The external pressure acting on each flow path is derived from the wind pressure, given by:

$$p_w = \rho / 2 C_p U^2 \text{ (Pa)}$$

and the stack pressure, given by:

$$p_s = -\rho_0 g 273 h (1/t_e - 1/t_i) \text{ (Pa)}$$

where  $\rho$  = air density ( $\text{kg/m}^3$ )  
 $\rho_0$  = air density at 273 K  
 $C_p$  = wind pressure coefficient  
 $h$  = height of opening (m)  
 $U$  = building height windspeed (m/s)  
 $t_e$  = external air temperature (K)  
 $t_i$  = internal air temperature (K)  
 $g$  = acceleration due to gravity ( $\text{m/s}^2$ )

The user defines up to 10 flow paths and enters appropriate flow characteristics, drawing data from knowl-

edge about the building or from the default data presented in Chapter 6 of the AIVC's Calculation Techniques Guide.

## 2 Program Operation

AIDA is written in BASIC, originally for a CASIO FX730P Pocket Computer. A full listing of this version is presented in Table 1. With minor modifications, however, it will also run on an IBM compatible PC, using either BASIC, GWBASIC or QUICK BASIC. Furthermore, if the QUICK BASIC compiler option is used, convergence is almost instant, and hence this algorithm may prove to be useful as part of a general energy or thermal heat loss model. Suggested amendments to the code for "PC" compatibility are given in Table 2. Initiation of the code will be machine dependent but in the BASIC environment, will normally be achieved by using the "RUN" command. Once the response "Welcome to AIDA" appears on the screen, then the "EXE" or "ENTER" key is pressed sequentially with the user responding accurately to each of the input questions.

Data entry is largely self explicit. The order of entry is:

- Building Volume ( $\text{m}^3$ )
  - Number of Flow Paths
  - Height of Flow Path (m)
  - Flow Coefficient ( $\text{m}^3/\text{s}$  at 1 Pa)
  - Flow Exponent
  - Wind Pressure Coefficient
  - External Temperature ( $^{\circ}\text{C}$ )
  - Internal Temperature ( $^{\circ}\text{C}$ )
  - Wind Speed
  - at Building Height (m/s)
- ) These items are repeated for each flow path.  
 ) necessary.

On completion of data entry, the computer responds with the message "Calculation in Progress". The calculation will take between < 1 second for a PC to > 60 seconds for the CASIO. After iteration is completed, the infiltration rate is displayed on the screen. If the CASIO machine is used, press "EXE" to display the air change rate and "EXE" again to enter further climatic data, break out of the program by pressing "AC". If the PC version is used, the air change rate and request for further climatic data is automatically displayed, break out of the program by using "CTRL BREAK".

Table 1: AIDA Program Listing for CASIO FX730P

```

10 SET N
20 PRINT "Welcome to AIDA"
30 PRINT "Air Infiltration Development Algorithm"
40 PRINT "M Liddament - AIVC -1989"
50 DIMH(10),C(10),N(10),P(10),T(10),W(10),S(10),F(10)
60 D = 1.29 : REM Air Density at 0 Deg C
70 PRINT "Enter Building Data:"
80 INPUT "Building Volume (m3) = ",V
90 PRINT "Enter Flow Path Data:"
100 INPUT "Number of Flow Paths = ",L
110 FOR J = 1 TO L
120 PRINT "Height (m)(Path";J;") = ";; INPUT H(J)
130 PRINT "Flow Coef (Path";J;") = ";; INPUT C(J)
140 PRINT "Flow Exp (Path";J;") = ";; INPUT N(J)
150 PRINT "Pres Coef (Path";J;") = ";; INPUT P(J)
160 NEXT J
170 PRINT "Enter Climatic Data:"
180 INPUT "Ext Temp (Deg C) = ",E
190 INPUT "Int Temp (Deg C) = ",I
200 INPUT "Wind Spd(Bldg Ht)(m/s) = ",U
210 REM Pressure Calculation
220 FOR J = 1 TO L
230 REM Wind Pressure Calculation
240 W(J) = .5*D*P(J)*U*U
250 REM Stack Pressure Calculation
260 S(J) = -3455*H(J)*(1/(E+273))-1/(I+273))
270 REM Total Pressure
280 T(J) = W(J) + S(J)
290 NEXT J
300 REM Calculate Infiltration
310 PRINT "Calculation in Progress";
320 R = -100
330 X = 50
340 Y = 0
350 B = 0
360 R = R + X
370 FOR J = 1 TO L
380 Y = Y + 1
390 O = T(J) - R
400 IF O = 0 THEN F(J) = 0: GOTO 430
410 F(J) = C(J)*(ABS(O) ^ N(J))*O/ABS(O)
420 B = B + F(J)
430 NEXT J
440 IF B < 0 THEN R = R - X: X = X/2: GOTO 350
450 IF B < 0.0001 THEN U GOTO 470
460 GOTO 350
470 Q = 0
480 FOR J = 1 TO L
490 IF F(J) > 0 THEN Q = Q + F(J)
500 NEXT J
510 SET F5
520 PRINT: PRINT "infiltration rate (m³/s) = ";Q
530 A = Q*3600/V
540 PRINT "air change rate (ach) = ";A
550 GOTO 170

```

At the completion of a session, the most recent data remains in store and can be recovered by entering the appropriate variable letter (CASIO) or by using the "PRINT" command (PC version).

Examples: (CASIO)

press "Q" followed by "EXE" to display infiltration rate  
 press "F(2)" followed by "EXE" to display flow in path 2

(PC version)

PRINT Q displays infiltration rate  
 PRINT F(2) displays flow in path 2

Table 2: Amendments for MSDOS (IBM Compatible) Operation

```

10 REM SET N
15 CLS
55 PRINT:PRINT:PRINT
85 PRINT:PRINT:PRINT
115 PRINT:PRINT:PRINT
165 PRINT:PRINT:PRINT
175 PRINT:PRINT:PRINT
305 CLS:PRINT:PRINT:PRINT
505 PRINT:PRINT:PRINT
510 REM SET F5
545 PRINT:PRINT:PRINT

```

Note: These amendments remove the CASIO "SET" commands and improve the interactive display.

All the variables used in the algorithm are listed in Table 3.

Data must be inserted with care since there is no error trapping or editing facility. If an error is made, then it is necessary to restart the program.

Clearly, since this is primarily a pocket computer algorithm, the interactive Input/Output routine is very rudimentary. It is assumed that if wider use is made of the "PC" version, then the user will amend these routines to suit individual requirements.

### 3 Solution Technique.

The flow balance equation is solved iteratively using a combination of "bi-section" and "addition". An internal pressure, known to be substantially negative with respect to the true pressure, is selected as a starting condition. For most applications a value of -100 Pa should be satisfactory and is automatically introduced at line 320. Successive iterations improve upon the internal pressure value until a flow balance within  $0.0001 \text{ m}^3/\text{s}$  is

Table 3: AIDA List of Variables

A	= Air Change Rate (ACH)
B	= Flow Balance
C(J)	= Flow Coefficient (Path J)
D	= Air Density ( $\text{kg}/\text{m}^3$ )
E	= External Temperature (Degrees C)
F(J)	= Calculated Flow Rate (Path J) ( $\text{m}^3/\text{s}$ )
H(J)	= Height of Flow Path (Path J) (m)
I	= Internal Temperature (Degrees C)
J	= Flow Path Number
L	= Total Number of Flow Paths (Max = 10)
N(J)	= Flow Exponent (Path J)
O	= Pressure Difference across flow path (Pa)
P(J)	= Wind Pressure Coefficient (Path J)
Q	= Infiltration Rate ( $\text{m}^3/\text{s}$ )
R	= Internal Pressure (Pa)
S(J)	= Stack Induced Pressure (Path J) (Pa)
T(J)	= Total External Pressure on Flow Path (Path J) (Pa)
U	= Wind Speed at Building Height (m/s)
V	= Volume of Building or Enclosure ( $\text{m}^3$ )
W(J)	= Wind Induced Pressure (Path J) (Pa)
X	= Iteration Pressure Step (Pa)
Y	= Iteration Counter

achieved. The flow balance criterion is established in line 450. An understanding of the technique may be gleaned from an analysis of lines 320 to 470 of the program. While this approach may not necessarily be the most numerically efficient, it is extremely robust and should be applicable over a wide range of flow conditions and leakage characteristics.

#### 4 Example

Input data and results for a 3 flow path network are presented in Table 4. Although this is an arbitrary example, it has been selected to show that AIDA can handle differing flow coefficients, flow exponents, flow path heights and wind pressure coefficients within the same network.

#### 5 Program Developments

Developments covering mechanical ventilation, additional flow paths, automatic wind pressure distributions and wind speed correction equations may be readily incorporated. It is intended to cover some of these developments, along with validation tests in future issues of "AIR".

#### 6 References

Liddament, M.W. Air Infiltration Calculation Techniques - An Applications Guide. Air Infiltration and Ventilation Centre 1986.

Table 4: Example Data and Results

Building Volume = 250 m <sup>3</sup>			
Number of flow paths = 3			
	Path 1	Path 2	Path 3
Height of Path (m)	2	4	7
Flow Coef (m <sup>3</sup> /s at 1 pa)	0.03	0.06	0.02
Flow Exponent	0.7	0.5	0.6
Wind Pressure Coef	0.3	-0.25	-0.4
Climatic Data			
	Ext Temp (Deg C)	Int Temp (Deg C)	Wind Speed (m/s)
Run 1	0	20	0
Run 2	5	15	2
Run 3	18	18	3

Results:

	Infilt Rate (m <sup>3</sup> /s)	Air Ch Rate (ach)	Flow (Path 1) (m <sup>3</sup> /s)	Flow (Path 2) (m <sup>3</sup> /s)	Flow (Path 3) (m <sup>3</sup> /s)
Run 1	0.0437	0.63	0.0437	-0.0081	-0.0356
Run 2	0.0511	0.74	0.0511	-0.0224	-0.0287
Run 3	0.0615	0.89	0.0615	-0.0382	-0.0232

## Final Reminder - AIVC Survey of Research

Replies are still being accepted for the AIVC Survey of Current Research into Air Infiltration and Ventilation 1989. Your completed survey form will be entered on a database and published in the form of an AIVC Technical Note.

The Technical Note is automatically distributed to all researchers listed, thus guaranteeing free publicity for your particular project in all thirteen participating countries of the AIVC.

The Survey database will be available for distribution on Archivist software, as part of the AIVC "Airbase" package of diskettes.

Any outstanding Survey Forms should be completed and returned, preferably by telefax, to Mark Limb at AIVC as soon as possible.